
Machine-level  Semantics
Interpreters written in a clean functional metalanguage (like 
functional Scheme or Haskell) provide clear definitions of 
meaning for programming languages.  We can even tolerate 
occasional use of imperative features (e.g., mutation operations 
on shared data in Scheme) and still claim that interpreters clearly 
define the behavior of programs (provided we define the 
semantics of our imperative extension of metalanguage).
But these interpreters do not describe how to implement 
programming languages in terms of conventional machine 
instructions.  What is missing?  A description of the meaning of 
function calls and error operations used in the interpreter 
metalanguage.  In principle we would like to know how to hand 
translate our interpreters to machine code, or even better, how to 
compile source programs directly into machine code.



Guidance From an Example
Consider the following procedure, which computes the product of a list 
of numbers:

(define Pi
  (lambda (l)
    (cond
      ((null? l) 1)
      (else (* (first l) (Pi (rest l)))))))

What if the input list may be corrupt (contain non-numbers)?

(define Pi-acc 
  (lambda (l acc)
    (cond ((null? l) acc)
          ((number? (first l)) (Pi-acc (rest l) (* (first l) acc)))
          (else (first l)))))  ;; return erroneous input

(define Pi-2 (lambda (l) (Pi-acc l 1))))

Does Pi-2 preserve the order of evaluation in Pi?



Guidance From an Example cont.

Suppose Pi-2 is passed a corrupt list. In that case, Pi-acc multiplies all 
the numbers found until the erroneous input is encountered. Can we 
avoid these wasted multiplications?  Yes.  By making acc into a 
suspension.

(define Pi-acc
   (lambda (l acc)
     (cond ((null? l) (acc))
           ((number? (first l)) 
            (Pi/acc (rest l) (lambda () (* (first l) (acc)))))
           (else (first l)))))

(define Pi-3 (lambda (l) (Pi/acc l (lambda () 1)))))

This program avoids unnecessary multiplications.  But like  Pi-2, it 
changes the order of multiplications from Pi.  If the * primitive were 
not associative, the transformation used to create Pi-2 would not 
work.



Systematically Avoiding Nested Function Calls
We failed to preserve the evaluation order in Pi in constructing Pi-2 and 
Pi-3 because updating an accumulator  reverses the order of operations on 
a list.  Is there a systematic way to avoid nesting function calls while 
preserving evaluation order?  Yes!  It is called transformation to 
continuation-passing style (CPS).  The CPS transform of Pi is:

(define Pi-k
  (lambda (l k)                 ;; k is the rest of the computation
    (cond ((null? l) (k 1))
          ((number? (first l)) 
           (Pi-k (rest l) (lambda (prod) (k (* (first l) prod)))))
          (else (first l)))))   ;; escape on an illegal input!

(define Pi-4 (lambda (l) (Pi-k l (lambda (x) x))))

Why does the (else  ... ) clause work? It is because Pi-4 is tail-recursive; 
it is only called at top-level.  Hence, the value returned by the else clause is 
guaranteed to return directly to the caller of Pi-4.  CPS converts all 
functions to tail-form (no nested function calls).



The CPS Transformation
Assume Jam/Scheme programs are restricted to a form where the body of a function is either 
(i) a primitive expression constructed from constants, variables and primitive functions,  and 
program-defined functions; or
(ii) a conditional where the predicates are primitive expressions and the result clauses are 
ordinary expressions (primitive expressions augmented by program-defined functions) .  Then 
the CPS transformation of such a program is defined as follows:

1.  Add an extra parameter k to every function.

2.  For each function body b that is a primitive expression, write (k b).

a)3.  Each clause in a conditional is treated separately:

a) For each result clause b composed from primitive operations and constants, write      
(k b).

b) For each clause containing calls on program-defined functions, pick the call that will 
be evaluated first. Make the body for the new clause a call that takes an extra 
argument, which is of the form (lambda (res) body).  The original contents of that 
clause are placed in the body, enclosed in a call on the continuation k, with the 
selected recursive call replaced by res.

– Repeat step (3b) until no unconverted function calls remain.



Another Example
(define Pi
  (lambda (t)
    (cond
      ((leaf? t) t)
      (else (* (Pi (left t))
               (Pi (right t)))))))

Then first iteration in creating the CPS version, Pi-k, is
(define Pi-k
  (lambda (t k)                      ;; rule 1
    (cond ((leaf? t) (k t))          ;; rule 3a
          (else                      ;; rule 3b
            (Pi-k (left t)
                  (lambda (res)
                    (k (* res (Pi (right t))))))))))



Second Iteration

 (define Pi-k
   (lambda (t k)                  ; rule 1
     (cond ((leaf? t) (k t))      ; rule 3a
           (else                  ; rule 3b
             (Pi-k (left t)
                   (lambda (r1)
                     (Pi-k (right t)
                           (lambda (r2) (k (* r1 r2))))))))))

